Experimental Mathematics and Data Mining:

Excavating the Online Encyclopedia of Integer Sequences

Hieu D. Nguyen

February 23, 2011

```
<<"c:/documents and settings/nguyen/my documents/2011/math/projects/experimental m
```

```
Mathematics by Experiment Package:
Implementation by Hieu Nguyen accompanying the textbook
 "Mathematics by Experiment: Exploring Patterns of Integer Sequences"
    -- Rowan University -- Version 1.0 (2/9/2011)
```

General::compat :

Combinatorica Graph and Permutations functionality has been superseded by preloaded functionaliy. The package now being loaded may conflict with this. Please see the Compatibility Guide for details.

Exploring Patterns of Integer Sequences

What's the Pattern?

Example - Pattern Recognition

1. Counting Rabbits

Consider the finite sequence {0, 1, 1, 2, 3, 5, 8, 13}.

a) What's the next term?

Next Term

21

b) What's the recurrence?

```
? FindLinearRecurrence
```

FindLinearRecurrence[*list*] finds if possible the minimal linear recurrence that generates *list*.
FindLinearRecurrence[*list*, *d*] finds if possible the linear recurrence of maximum order *d* that generates *list*. ≫

```
FindLinearRecurrence[{1, 1, 2, 3, 5, 8, 13}]
{1, 1}
```

Recurrence

$F(n+1) = F(n) + F(n-1)$

c) What's the formula?

```
? FindSequenceFunction
```

FindSequenceFunction[{a_1, a_2, a_3, …}] attempts to find a simple function that yields the sequence a_n when given successive integer arguments.
FindSequenceFunction[{{n_1, a_1}, {n_2, a_2}, …}] attempts to find a simple function that yields a_i when given argument n_i.
FindSequenceFunction[*list*, *n*] gives the function applied to *n*. ≫

```
FindSequenceFunction[{1, 1, 2, 3, 5, 8, 13}, n]
Fibonacci[n]
```

```
FindSequenceFunction[{0, 1, 1, 2, 3, 5, 8, 13}, n]
```

$\frac{1}{2}$ (-Fibonacci[n] + LucasL[n])

2. Partial Sums

Consider the partial sums of the Fibonacci sequence: {0, 0 + 1, 0 + 1 + 1, 0 + 1 + 1 + 2, 0 + 1 + 1 + 2 + 3, ...}

```
Prepend[Table[
  {n, Fibonacci[n], If[n < 6, Sum[Fibonacci[k], {k, 0, n}], If[n == 6, "?", "-"]]},
  {n, 0, 8}], {"n", "F(n)", "∑_{k=0}^{n} F(k)"}] // Grid
```

n	F(n)	$\sum_{k=0}^{n} F(k)$
0	0	0
1	1	1
2	1	2
3	2	4
4	3	7
5	5	12
6	8	?
7	13	-
8	21	-

a) What's the next term?

Next Term

20

b) What's the formula?

```
Sum[Fibonacci[k], {k, 1, n}]
```

-1 + Fibonacci[2 + n]

Identity

$\sum_{k=0}^{n} F(k) = F(n+2) - 1$

NOTE: Applying the **FindSequenceFunction** yields a different formula:

```
Table[Sum[Fibonacci[k], {k, 0, n}], {n, 1, 10}]
```

{1, 2, 4, 7, 12, 20, 33, 54, 88, 143}

```
FindSequenceFunction[{1, 2, 4, 7, 12, 20, 33, 54, 88, 143}, n]
```

$\frac{1}{2}$ (-2 + 3 Fibonacci[n] + LucasL[n])

Equating the two formulas produces the following identity:

Identity

$F(n+2)-1=(3\,F(n)+L(n)-2)/2$
$\therefore\ L(n)=2\,F(n+2)-3\,F(n)$

```
Table[{LucasL[n], 2 Fibonacci[n + 2] - 3 Fibonacci[n]}, {n, 1, 10}]
```

c) What's the recurrence?

```
FindLinearRecurrence[{1, 2, 4, 7, 12, 20, 33, 54, 88, 143}]
```

```
{2, 0, -1}
```

Recurrence

(1) $a(n)=\sum_{k=0}^{n} F(k)$
(2) $a(n)=2\,a(n-1)-a(n-3)$

PROOF:

1. Substitute (1) into (2) and reduce (cancel summations):

```
Clear[a];
a[n_] := Sum[Fibonacci[k], {k, 0, n}]
reduce = Simplify[a[n] == 2 a[n - 1] - a[n - 3]]
```

```
Fibonacci[-1 + n] + Fibonacci[2 + n] == 2 Fibonacci[1 + n]
```

2. Apply Fibonacci recurrence and simplify:

```
Simplify[reduce /. Fibonacci[2 + n] -> Fibonacci[1 + n] + Fibonacci[n]]
```

```
Fibonacci[-1 + n] + Fibonacci[n] == Fibonacci[1 + n]
```

- **3. Sums of Squares**

Consider sums of squares of Fibonacci numbers: $\{0^2,\ 0^2+1^2,\ 0^2+1^2+1^2,\ 0^2+1^2+1^2+2^2,\ \ldots\}$

```
Prepend[Table[
  {n, Fibonacci[n], If[n < 6, Sum[Fibonacci[k]^2, {k, 0, n}], If[n == 6, "?", "-"]]},
  {n, 0, 8}], {"n", "F(n)", "∑_{k=0}^n F(k)^2"}] // Grid
```

n	F(n)	$\sum_{k=0}^{n} F(k)^2$
0	0	0
1	1	1
2	1	2
3	2	6
4	3	15
5	5	40
6	8	?
7	13	-
8	21	-

a) What's the next term?

Next Term

104

b) What's the formula?

```
Sum[Fibonacci[k]^2, {k, 0, n}]
```

```
Fibonacci[n] Fibonacci[1 + n]
```

Formula

$\sum_{k=0}^{n} F(k)^2 = F(n)\,F(n+1)$

NOTE: Again the **FindSequenceFunction** yields a different formula:

```
Sum[Fibonacci[k]^2, {k, 0, #}] & /@ Range[1, 10]
```

```
{1, 2, 6, 15, 40, 104, 273, 714, 1870, 4895}
```

```
FindSequenceFunction[{1, 2, 6, 15, 40, 104, 273, 714, 1870, 4895}, n]
```

$$-\frac{1}{10\left(5+2\sqrt{5}\right)}\left(10\,(-1)^n + 4\,(-1)^n\sqrt{5} + 5\left(\frac{3}{2}-\frac{\sqrt{5}}{2}\right)^n + 3\sqrt{5}\left(\frac{3}{2}-\frac{\sqrt{5}}{2}\right)^n - 15\left(\frac{3}{2}+\frac{\sqrt{5}}{2}\right)^n - 7\sqrt{5}\left(\frac{3}{2}+\frac{\sqrt{5}}{2}\right)^n\right)$$

c) What's the recurrence?

```
FindLinearRecurrence[{1, 2, 6, 15, 40, 104, 273, 714, 1870, 4895}]
```

```
{2, 2, -1}
```

Recurrence

$a(n) = \sum_{k=0}^{n} F(k)^2$
$a(n) = 2\,a(n-1) + 2\,a(n-2) - a(n-3)$

Example - Fibonacci's Cousin

1. Lucas Sequence

The Lucas sequence is defined by the recurrence $L(n+1) = L(n) + L(n-1)$ with $L(0) = 2$ and $L(1) = 1$.

```
Table[LucasL[n], {n, 0, 7}]
```

```
{2, 1, 3, 4, 7, 11, 18, 29}
```

```
FindSequenceFunction[{1, 3, 4, 7, 11, 18, 29}, n]
```

```
LucasL[n]
```

```
FindSequenceFunction[{2, 1, 3, 4, 7, 11, 18, 29}, n]
```

$$\frac{1}{2} (5\,\text{Fibonacci}[n] - \text{LucasL}[n])$$

Identity

$L(n-1) = (5\,F(n) - L(n))/2$

$\therefore F(n) = (2\,L(n-1) + L(n))/5$

```
Table[{Fibonacci[n], (2 LucasL[n - 1] + LucasL[n]) / 5}, {n, 0, 10}]
```

```
{{0, 0}, {1, 1}, {1, 1}, {2, 2}, {3, 3},
 {5, 5}, {8, 8}, {13, 13}, {21, 21}, {34, 34}, {55, 55}}
```

2. Partial Sums

Consider the partial sums of the Lucas sequence:

```
Prepend[Table[{n, LucasL[n], Sum[LucasL[k], {k, 0, n}]}, {n, 0, 5}],
  {"n", "L(n)", "∑_{k=0}^{n} L(k)"}] // Grid
```

n	L(n)	$\sum_{k=0}^{n} L(k)$
0	2	2
1	1	3
2	3	6
3	4	10
4	7	17
5	11	28

a) What's the next term?

Next term

46

b) What’s the recurrence?

Recurrence

$b(n) = \sum_{k=0}^{n} L(k)$
$b(n) = b(n-1) + b(n-2) + 1$

```
FindLinearRecurrence[{2, 3, 6, 10, 17, 28, 46, 75, 122, 198, 321}]
```

```
{2, 0, -1}
```

Recurrence

$b(n) = \sum_{k=0}^{n} L(k)$
$b(n) = 2\, b(n-1) - b(n-3)$

c) What’s the formula?

```
Sum[LucasL[k], {k, 0, n}]
```

$$-\left(-1-\sqrt{5}\right)^{-1-n} \left(\left(-1-\sqrt{5}\right)^{1+n} + 2^n \left(-1+\sqrt{5}\right) + 2 \left(-3-\sqrt{5}\right)^n \left(2+\sqrt{5}\right)\right)$$

```
Table[Sum[LucasL[k], {k, 0, n}], {n, 0, 10}]
```

```
{2, 3, 6, 10, 17, 28, 46, 75, 122, 198, 321}
```

```
FindSequenceFunction[{3, 6, 10, 17, 28, 46, 75, 122, 198, 321}, n]
```

$$\frac{1}{2} \left(-2 + 5\,\text{Fibonacci}[n] + 3\,\text{LucasL}[n]\right)$$

Formula

$\sum_{k=0}^{n} L(k) = (5\,F(n) + 3\,L(n) - 2)/2$

NOTE: Recall that $\sum_{k=0}^{n} F(k) = (3\,F(n) + L(n) - 2)/2$. Subtracting these two formulas yields the identity

$\sum_{k=0}^{n} [L(k) - F(k)] = F(n) + L(n)$

```
Table[{Sum[LucasL[k] - Fibonacci[k], {k, 0, n}], Fibonacci[n] + LucasL[n]}, {n, 0, 10}]
```

```
{{2, 2}, {2, 2}, {4, 4}, {6, 6}, {10, 10}, {16, 16},
 {26, 26}, {42, 42}, {68, 68}, {110, 110}, {178, 178}}
```

■ 3. Binomial Convolution

$b(n) = \sum_{k=0}^{n} \binom{n}{k} a(k)$

Consider the binomial convolution of the Lucas sequence:
1·2

$1 \cdot 2 + 1 \cdot 1$
$1 \cdot 2 + 2 \cdot 1 + 1 \cdot 3$
$1 \cdot 2 + 3 \cdot 1 + 3 \cdot 3 + 1 \cdot 4$
...

```
Prepend[Table[{n, LucasL[n],
    If[n < 5, Sum[Binomial[n, k] * LucasL[k], {k, 0, n}], If[n == 5, "?", "-"]]},
   {n, 0, 10}], {"n", "L(n)", "∑_{k=0}^{n} (n k) L(n)"}] // Grid
```

n	L(n)	$\sum_{k=0}^{n} \binom{n}{k} L(n)$
0	2	2
1	1	3
2	3	7
3	4	18
4	7	47
5	11	?
6	18	-
7	29	-
8	47	-
9	76	-
10	123	-

a) What’s the next term?

Next term

123

b) What’s the formula?

```
Sum[Binomial[n, k] * LucasL[k], {k, 0, n}]
```

$$\left(\frac{1}{2}\left(3-\sqrt{5}\right)\right)^n + \left(\frac{1}{2}\left(3+\sqrt{5}\right)\right)^n$$

```
tempdata = Table[Sum[Binomial[n, k] * LucasL[k], {k, 0, n}], {n, 1, 7}]
```

{3, 7, 18, 47, 123, 322, 843}

```
FindSequenceFunction[tempdata, n]
```

$$\left(\frac{3}{2}-\frac{\sqrt{5}}{2}\right)^n + \left(\frac{3}{2}+\frac{\sqrt{5}}{2}\right)^n$$

c) What’s the recurrence?

```
FindLinearRecurrence[tempdata]
```

{3, -1}

Online Encyclopedia of Integer Sequences (OEIS)

OEIS Web Site: http://oeis.org/

- **Searchable database containing over 180,000 entries**

```
Hyperlink[Style["OEIS Web Site", Plain],
 "http://oeis.org", Appearance -> "DialogBox"]
```

OEIS Web Site

Example - Fibonacci’s Cousin (continued)

- **3. Binomial Sum (continued)**

```
tempdata = Table[Sum[Binomial[n, k] * LucasL[k], {k, 0, n}], {n, 1, 7}]
```

{3, 7, 18, 47, 123, 322, 843}

```
Hyperlink[Style["OEIS Search Results", Plain], "http://oeis.org/search?q=" <>
  ToString[tempdata] <> "&language=english&go=Search", Appearance -> "DialogBox"]
```

OEIS Search Results

Formula

$$\sum_{k=0}^{n} \binom{n}{k} L(k) = L(2\,n)$$

- **4. Sums of Squares of Odd Terms**

```
tempdata = Table[Sum[LucasL[2 k - 1]^2, {k, 1, n}], {n, 0, 10}]
```

{0, 1, 17, 138, 979, 6755, 46 356, 317 797, 2 178 293, 14 930 334, 102 334 135}

What’s the formula?

```
Sum[LucasL[2 k - 1]^2, {k, 1, n}]
```

$$-\frac{1}{5+3\sqrt{5}}\left(2\left(-1-\sqrt{5}\right)\right)^{-4n}\Big(3\times 2^{8n}+2^{8n}\sqrt{5}-5\times 2^{1+4n}\left(-1-\sqrt{5}\right)^{4n}-3\times 2^{1+4n}\sqrt{5}\left(-1-\sqrt{5}\right)^{4n}+7\times 2^{1+4n}\left(\left(-1-\sqrt{5}\right)\left(1+\sqrt{5}\right)\right)^{2n}-2^{2+4n}\left(\left(-1-\sqrt{5}\right)\left(1+\sqrt{5}\right)\right)^{2n}+3\times 2^{1+4n}\sqrt{5}\left(\left(-1-\sqrt{5}\right)\left(1+\sqrt{5}\right)\right)^{2n}-3\left(\left(-1-\sqrt{5}\right)\left(1+\sqrt{5}\right)\right)^{4n}-\sqrt{5}\left(\left(-1-\sqrt{5}\right)\left(1+\sqrt{5}\right)\right)^{4n}+7\times 2^{1+4n}\left(\left(-1-\sqrt{5}\right)\left(1+\sqrt{5}\right)\right)^{2n}n-2^{2+4n}\left(\left(-1-\sqrt{5}\right)\left(1+\sqrt{5}\right)\right)^{2n}n+3\times 2^{1+4n}\sqrt{5}\left(\left(-1-\sqrt{5}\right)\left(1+\sqrt{5}\right)\right)^{2n}n\Big)$$

```
Simplify[%]
```

$$\frac{1}{5+3\sqrt{5}}\left(-2\left(1+\sqrt{5}\right)\right)^{-4n}\Big((-2)^{4n}\left(3+\sqrt{5}\right)^{1+4n}-16^{n}\Big(16^{n}\left(3+\sqrt{5}\right)+2^{1+2n}\left(-3-\sqrt{5}\right)^{2n}\left(5+3\sqrt{5}\right)-2\left(-1-\sqrt{5}\right)^{4n}\left(5+3\sqrt{5}\right)\Big)-2^{1+6n}\left(-3-\sqrt{5}\right)^{2n}\left(5+3\sqrt{5}\right)n\Big)$$

```
FindSequenceFunction[Delete[tempdata, 1], n]
```

$$-\Bigg(5\times 2^{4-n}\Bigg(13\,997\,205\times 2^{3+n}\left(\frac{7}{2}-\frac{3\sqrt{5}}{2}\right)^{n}+50\,077\,923\sqrt{5}\left(2\left(\frac{7}{2}-\frac{3\sqrt{5}}{2}\right)\right)^{n}-13\,997\,205\times 2^{3+n}\left(\frac{7}{2}+\frac{3\sqrt{5}}{2}\right)^{n}-50\,077\,923\sqrt{5}\left(2\left(\frac{7}{2}+\frac{3\sqrt{5}}{2}\right)\right)^{n}+74\,651\,760\left(\left(7-3\sqrt{5}\right)\left(\frac{7}{2}+\frac{3\sqrt{5}}{2}\right)\right)^{n}+33\,385\,282\sqrt{5}\left(\left(7-3\sqrt{5}\right)\left(\frac{7}{2}+\frac{3\sqrt{5}}{2}\right)\right)^{n}-74\,651\,760\left(\left(\frac{7}{2}-\frac{3\sqrt{5}}{2}\right)\left(7+3\sqrt{5}\right)\right)^{n}-33\,385\,282\sqrt{5}\left(\left(\frac{7}{2}-\frac{3\sqrt{5}}{2}\right)\left(7+3\sqrt{5}\right)\right)^{n}+63\,760\,215\left(\left(7-3\sqrt{5}\right)\left(\frac{7}{2}+\frac{3\sqrt{5}}{2}\right)\right)^{n}n+28\,514\,435\sqrt{5}\left(\left(7-3\sqrt{5}\right)\left(\frac{7}{2}+\frac{3\sqrt{5}}{2}\right)\right)^{n}n+437\,019\,015\left(\left(\frac{7}{2}-\frac{3\sqrt{5}}{2}\right)\left(7+3\sqrt{5}\right)\right)^{n}n+195\,440\,845\sqrt{5}\left(\left(\frac{7}{2}-\frac{3\sqrt{5}}{2}\right)\left(7+3\sqrt{5}\right)\right)^{n}n\Bigg)\Bigg)\Bigg/\left(3\left(-5+3\sqrt{5}\right)^{2}\left(5+3\sqrt{5}\right)^{2}\left(16\,692\,641+7\,465\,176\sqrt{5}\right)\right)$$

```
Hyperlink[Style["OEIS Search Results", Plain], "http://oeis.org/search?q=" <>
  ToString[tempdata] <> "&language=english&go=Search", Appearance -> "DialogBox"]
```

OEIS Search Results

Formula

$\sum_{k=1}^{n} L(2\,k-1)^2 = F(4\,n) - 2\,n$

Generating Recursive Sequences in *Mathematica* Efficiently

Example - A003501

Consider the sequence

$a(n) = 5\,a(n-1) - a(n-2);\ a(0) = 2,\ a(1) = 5$

Here are five methods for generating $a(n)$:

METHOD 1

```
Clear[a];
a[0] = 2;
a[1] = 5;
a[n_] := 5 a[n - 1] - a[n - 2]
Table[a[n], {n, 0, 10}]
```

```
{2, 5, 23, 110, 527, 2525, 12 098, 57 965, 277 727, 1 330 670, 6 375 623}
```

```
Timing[Table[a[n], {n, 0, 30}];]
```

```
{14.703, Null}
```

METHOD 2

```
Clear[a];
a[0] = 2;
a[1] = 5;
a[n_] := a[n] = 5 a[n - 1] - a[n - 2]
Table[a[n], {n, 0, 10}]
```

```
{2, 5, 23, 110, 527, 2525, 12 098, 57 965, 277 727, 1 330 670, 6 375 623}
```

```
Timing[Table[a[n], {n, 0, 30}];]
```

```
{0., Null}
```

- **METHOD 3**

```
Clear[a];
a[0] = 2;
a[1] = 5;
sequence[nMax_] := Module[{n},
   Do[
    a[n] = 5 a[n - 1] - a[n - 2],
    {n, 2, nMax}
   ];
   Table[a[n], {n, 0, nMax}]
  ]
```

```
Timing[sequence[30];]
```

```
{0., Null}
```

- **METHOD 4**

```
? LinearRecurrence
```

LinearRecurrence[*ker*, *init*, *n*] gives the sequence of length *n*
obtained by iterating the linear recurrence with kernel *ker* starting with initial values *init*.
LinearRecurrence[*ker*, *init*, {n_{min}, n_{max}}] yields terms n_{min} through n_{max} in the linear recurrence sequence. ≫

```
LinearRecurrence[{5, -1}, {2, 5}, 10]
```

```
{2, 5, 23, 110, 527, 2525, 12 098, 57 965, 277 727, 1 330 670}
```

```
Timing[LinearRecurrence[{5, -1}, {2, 5}, 30];]
```

```
{0.016, Null}
```

- **METHOD 5**

```
Clear[a];
a[n_] =
 FindSequenceFunction[{5, 23, 110, 527, 2525, 12 098, 57 965, 277 727, 1 330 670}, n]
```

$$\left(\frac{5}{2}-\frac{\sqrt{21}}{2}\right)^n+\left(\frac{5}{2}+\frac{\sqrt{21}}{2}\right)^n$$

```
Timing[Simplify[Table[a[n], {n, 0, 30}]];]
```

```
{0.203, Null}
```

■ Exponential Subsequence (Powers of 2)

How can we efficiently generate the terms of the exponential subsequence $a(2^n)$ for n ranging from 0 to 10?

■ METHOD 2

```
Clear[a];
a[0] = 2;
a[1] = 5;
a[n_] := a[n] = 5 a[n - 1] - a[n - 2]
```

```
tempdata = Table[a[2^n], {n, 0, 10}]
```

$RecursionLimit::reclim : Recursion depth of 256 exceeded. ≫

$RecursionLimit::reclim : Recursion depth of 256 exceeded. ≫

```
$Aborted
```

```
tempdata = Table[a[2^n], {n, 0, 8}]
```

```
{5, 23, 527, 277 727, 77 132 286 527, 5 949 389 624 883 225 721 727,
 35 395 236 908 668 169 265 765 137 996 816 180 039 862 527,
 1 252 822 795 820 745 419 377 249 396 736 955 608 088 527 968 701 950 139 470 082 687 906 021 \
  780 162 741 058 825 727,
 1 569 564 957 728 109 166 248 928 540 692 850 198 959 845 268 398 133 677 497 622 880 296 999 \
  933 490 617 154 569 622 109 395 310 689 810 853 415 707 068 113 663 529 488 212 649 183 417 \
  413 913 396 122 424 895 838 735 880 157 078 527}
```

■ Formula for exponential subsequence?

```
FindSequenceFunction[tempdata, n]
```

```
FindSequenceFunction[{5, 23, 527, 277 727, 77 132 286 527,
  5 949 389 624 883 225 721 727, 35 395 236 908 668 169 265 765 137 996 816 180 039 862 527,
  1 252 822 795 820 745 419 377 249 396 736 955 608 088 527 968 701 950 139 470 082 687 906 021 \
   780 162 741 058 825 727,
  1 569 564 957 728 109 166 248 928 540 692 850 198 959 845 268 398 133 677 497 622 880 296 999 \
   933 490 617 154 569 622 109 395 310 689 810 853 415 707 068 113 663 529 488 212 649 183 417 \
   413 913 396 122 424 895 838 735 880 157 078 527}, n]
```

■ Recursion for exponential subsequence?

```
FindLinearRecurrence[tempdata]
```

```
FindLinearRecurrence[{5, 23, 527, 277 727, 77 132 286 527,
  5 949 389 624 883 225 721 727, 35 395 236 908 668 169 265 765 137 996 816 180 039 862 527,
  1 252 822 795 820 745 419 377 249 396 736 955 608 088 527 968 701 950 139 470 082 687 906 021 \
   780 162 741 058 825 727,
  1 569 564 957 728 109 166 248 928 540 692 850 198 959 845 268 398 133 677 497 622 880 296 999 \
   933 490 617 154 569 622 109 395 310 689 810 853 415 707 068 113 663 529 488 212 649 183 417 \
   413 913 396 122 424 895 838 735 880 157 078 527}]
```

- **METHOD 3**

```
Clear[a];
a[0] = 2;
a[1] = 5;
exponentialsubsequence[nMax_] := Module[{n},
 Do[
  a[n] = 5 a[n - 1] - a[n - 2],
  {n, 2, 2^nMax}
  ];
 Table[a[2^n], {n, 0, nMax}]
 ]
```

```
Timing[exponentialsubsequence[10];]
```

{0.031, Null}

- **METHOD 4**

```
Timing[
 Part[LinearRecurrence[{5, -1}, {2, 5}, 2^10 + 1], #] & /@ Table[2^n + 1, {n, 0, 10}];]
```

{0.078, Null}

- **METHOD 5**

```
Clear[a];
a[n_] = Simplify[
  FindSequenceFunction[{5, 23, 110, 527, 2525, 12 098, 57 965, 277 727, 1 330 670}, n]]
```

$2^{-n} \left(\left(5-\sqrt{21}\right)^n+\left(5+\sqrt{21}\right)^n\right)$

```
a[2^n]
```

$2^{-2^n} \left(\left(5-\sqrt{21}\right)^{2^n}+\left(5+\sqrt{21}\right)^{2^n}\right)$

```
Timing[Simplify[Table[a[2^n], {n, 0, 10}]]]
```

```
{0.109, {5, 23, 527, 277 727, 77 132 286 527, 5 949 389 624 883 225 721 727,
  35 395 236 908 668 169 265 765 137 996 816 180 039 862 527,
  ((5 - √21)^128 + (5 + √21)^128) / 340 282 366 920 938 463 463 374 607 431 768 211 456,
  ((5 - √21)^256 + (5 + √21)^256) /
   115 792 089 237 316 195 423 570 985 008 687 907 853 269 984 665 640 564 039 457 584 007 913 ⋱
    129 639 936, ((5 - √21)^512 + (5 + √21)^512) /
   13 407 807 929 942 597 099 574 024 998 205 846 127 479 365 820 592 393 377 723 561 443 721 ⋱
    764 030 073 546 976 801 874 298 166 903 427 690 031 858 186 486 050 853 753 882 811 946 569 ⋱
    946 433 649 006 084 096, ((5 - √21)^1024 + (5 + √21)^1024) /
   179 769 313 486 231 590 772 930 519 078 902 473 361 797 697 894 230 657 273 430 081 157 732 ⋱
    675 805 500 963 132 708 477 322 407 536 021 120 113 879 871 393 357 658 789 768 814 416 622 ⋱
    492 847 430 639 474 124 377 767 893 424 865 485 276 302 219 601 246 094 119 453 082 952 085 ⋱
    005 768 838 150 682 342 462 881 473 913 110 540 827 237 163 350 510 684 586 298 239 947 245 ⋱
    938 479 716 304 835 356 329 624 224 137 216}}
```

- **OEIS Search**

```
Hyperlink[Style["OEIS Search Results", Plain],
 "http://oeis.org/search?q=" <> ToString[exponentialsubsequence[10]] <>
  "&language=english&go=Search", Appearance -> "DialogBox"]
```

```
OEIS Search Results
```

```
Hyperlink[Style["OEIS Search Results", Plain],
 "http://oeis.org/search?q=" <> ToString[exponentialsubsequence[5]] <>
  "&language=english&go=Search", Appearance -> "DialogBox"]
```

```
OEIS Search Results
```

EXPERIMENTAL CONJECTURE: Define $b(n) = a(2^n)$. Then $b(n)$ satisfies the non-linear recurrence

$b(n) = b(n-1)^2 - 2$

```
Clear[b, n];
b[0] = 5;
b[n_] := b[n] = b[n - 1]^2 - 2
```

```
Timing[Table[b[n], {n, 0, 10}]]
```

```
{1.17961 × 10^-15, {5, 23, 527, 277 727, 77 132 286 527,
  5 949 389 624 883 225 721 727, 35 395 236 908 668 169 265 765 137 996 816 180 039 862 527,
  1 252 822 795 820 745 419 377 249 396 736 955 608 088 527 968 701 950 139 470 082 687 906 021 ⋱
   780 162 741 058 825 727,
  1 569 564 957 728 109 166 248 928 540 692 850 198 959 845 268 398 133 677 497 622 880 296 999 ⋱
   933 490 617 154 569 622 109 395 310 689 810 853 415 707 068 113 663 529 488 212 649 183 417 ⋱
   413 913 396 122 424 895 838 735 880 157 078 527,
  2 463 534 156 528 041 113 959 753 710 513 002 205 852 603 826 266 586 277 940 048 183 964 976 ⋱
   303 955 490 951 944 411 520 604 181 024 106 041 218 857 920 444 236 026 200 238 532 878 283 ⋱
   725 775 526 848 105 251 113 420 758 158 080 676 005 003 093 972 361 031 748 858 494 820 780 ⋱
   190 473 207 480 427 244 716 252 068 776 250 150 520 634 002 565 000 828 386 761 431 028 437 ⋱
   771 364 436 656 983 082 889 499 645 917 229 620 923 899 105 012 328 523 169 571 183 644 489 ⋱
   727,
  6 069 000 540 380 326 976 303 110 768 424 892 037 373 923 411 431 254 232 065 781 482 843 702 ⋱
   615 754 398 854 501 287 703 168 842 975 059 997 594 303 796 864 638 608 271 088 777 955 233 ⋱
   421 370 904 455 987 520 105 743 409 417 357 668 881 537 792 228 331 502 340 049 528 551 880 ⋱
   048 634 459 571 366 574 394 326 031 007 162 386 215 279 418 667 208 181 698 121 585 647 184 ⋱
   121 798 853 609 404 504 475 003 401 181 716 557 046 537 357 368 283 350 050 570 307 597 651 ⋱
   557 892 020 253 882 056 172 939 222 458 977 924 542 352 775 647 924 104 921 360 294 312 608 ⋱
   249 043 739 613 384 215 147 566 717 682 668 036 065 904 839 028 567 176 355 542 371 100 981 ⋱
   508 024 760 118 011 850 446 205 942 267 832 655 574 579 900 290 773 404 547 485 780 694 178 ⋱
   711 356 103 772 903 146 327 547 519 168 688 504 781 529 330 317 402 989 198 453 963 895 975 ⋱
   988 085 189 265 118 159 657 464 494 036 869 639 527 418 139 687 167 759 702 292 841 090 208 ⋱
   534 527}}
```

```
Table[b[n], {n, 0, 10}] == exponentialsubsequence[10]
```

```
True
```

EXERCISE: Prove the conjecture above. Recall that

$$a(n) = 2^{-n}\left(\left(5-\sqrt{21}\right)^n + \left(5+\sqrt{21}\right)^n\right)$$

GENERALIZATION: Given a sequence $a(n)$ satisfying the linear recurrence $a(n) = c\, a(n) + d\, b(n)$, determine a recurrence for the exponential subsequence $b(n) = a(2^n)$.

Experimental Mathematics

What is Experimental Mathematics?

■ Jonathan Borwein and David Bailey

According to Borwein and Bailey [Mathematics by Experiment: Plausible Reasoning for the 21th Century, A K Peters, 2008], experimental mathematics is the methodology of doing mathematics that includes the use of computations for:

1. Gaining insight and intuition.

2. Discovering new patterns and relationships.

3. Using graphical displays to suggest underlying mathematical principles.

4. Testing and especially falsifying conjectures.

5. Exploring a possible result to see if it is worth formal proof.

6. Suggesting approaches for formal proof.

7. Replacing lengthy hand derivations with computer-based derivations.

8. Confirming analytically derived results.

Tools

Computer Algebra Systems (CAS)

- ***Mathematica***
- **Maple**
- **Matlab**
- **Sage**

Online Databases

- **Online Encyclopedia of Integer Sequences (OEIS): http://oeis.org - database of over 180,000 integer sequences**
- **Inverse Symbolic Calculator (ISC): http://oldweb.cecm.sfu.ca/projects/ISC/ISCmain.html - database of 54 million mathematical constants**

Algorithms

- **Generating functions**
- **Linear recurrences**
- **Partial Sums Least Squares (PSLQ) algorithm**
- **Gosper-Wilf-Zeilberger algorithms**

Data Mining

What is Data Mining?

- ### Large Scale Pattern Recognition

Data mining is the process of extracting patterns from large datasets using computer science, mathematics, and statistics.

■ Mining OEIS

■ Number Patterns of Integer Sequences



- **Integer Sequence Identities**



Example: Pell's Equation

Solutions to $x^2 - 2y^2 = \pm 1$

```
Pellsolutions = Sort[ FindInstance[(x^2 - 2 y^2 == -1 || x^2 - 2 y^2 == 1) &&
    0 < x < 250 && 0 < y < 250, {x, y}, Integers, 10]]
```

```
{{x → 1, y → 1}, {x → 3, y → 2}, {x → 7, y → 5},
 {x → 17, y → 12}, {x → 41, y → 29}, {x → 99, y → 70}, {x → 239, y → 169}}
```

```
dataPellsolutions = Table[{Pellsolutions[[k, 1, 2]], Pellsolutions[[k, 2, 2]]},
   {k, 1, Length[Pellsolutions]}];
Prepend[dataPellsolutions, {"x", "y"}] // Grid
```

x	y
1	1
3	2
7	5
17	12
41	29
99	70
239	169

Define $a(n) = x(n)\,y(n)$

```
tempdata = Table[dataPellsolutions[[k, 1]] * dataPellsolutions[[k, 2]],
  {k, 1, Length[dataPellsolutions]}]
```

```
{1, 6, 35, 204, 1189, 6930, 40 391}
```

Do you recognize a pattern?

Formula for $a(n)$

```
Clear[a];
a[n_] = FindSequenceFunction[tempdata, n]
```

$$-\left(\left(4+3\sqrt{2}\right)\left(\left(3-2\sqrt{2}\right)^n-\left(3+2\sqrt{2}\right)^n\right)\right)\Big/\left(8\left(3+2\sqrt{2}\right)\right)$$

```
FindLinearRecurrence[tempdata]
```

```
{6, -1}
```

```
Hyperlink[Style["OEIS Search", Plain], "http://oeis.org/search?q=" <>
  ToString[tempdata] <> "&language=english&go=Search", Appearance -> "DialogBox"]
```

OEIS Search

Identity involving *a*(*n*)

Transformation 1

```
Ta1[n_] := a[n] * a[n + 1]
```

```
tempdata1 = Simplify[Table[Ta1[n], {n, 0, 10}]]
```

```
{0, 6, 210, 7140, 242 556, 8 239 770, 279 909 630,
 9 508 687 656, 323 015 470 680, 10 973 017 315 470, 372 759 573 255 306}
```

```
Hyperlink[Style["OEIS Search", Plain], "http://oeis.org/search?q=" <>
  ToString[tempdata1] <> "&language=english&go=Search", Appearance -> "DialogBox"]
```

OEIS Search

MATCH: Ta1(n) = A029549

Transformation 2

```
Ta2[n_] := Sum[a[2 k], {k, 0, n}]
```

```
tempdata2 = Table[Simplify[Ta2[n]], {n, 0, 10}]
```

```
{0, 6, 210, 7140, 242 556, 8 239 770, 279 909 630,
 9 508 687 656, 323 015 470 680, 10 973 017 315 470, 372 759 573 255 306}
```

```
Hyperlink[Style["OEIS Search", Plain], "http://oeis.org/search?q=" <>
  ToString[tempdata2] <> "&language=english&go=Search", Appearance -> "DialogBox"]
```

OEIS Search

MATCH: Ta2(n) = A029549

Experimental Conjecture:

$\sum_{k=0}^{n} a(2\,k) = a(n)\,a(n+1)$

Mathematical Proof?

EUREKA Project

GOAL: Data mine the OEIS for number patterns (formulas and identities)

- **Computer automated search**
- ***Mathematica* implementation**

- **Why use a computer algebra system?**

Arbitrary large integers

Symbolic computation

- **Approach**

- **Save entire OEIS database as a text file (label, name, sequence, offset)**
- **Apply transformations to each integer sequence (or subsequence) in OEIS and search database text file for match**
- **Equate transformations which have the same unique match to generate an identity**

Programming Challenges

OEIS

1. Small number of terms for certain sequences (OEIS only requires a minimum of 4 terms)

2. Variations of the same sequence are listed; thus, many sequences have a significant number of terms in common:

Example - Zero Sequence

```
tempdata = Table[0, {n, 1, 30}]
```

```
{0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0}
```

```
Hyperlink[Style["OEIS Search", Plain], "http://oeis.org/search?q=" <>
  ToString[tempdata] <> "&language=english&go=Search", Appearance -> "DialogBox"]
```

```
OEIS Search
```

Example - Triangular Numbers

```
tempdata = Table[n (n + 1) / 2, {n, 0, 50}]
```

```
{0, 1, 3, 6, 10, 15, 21, 28, 36, 45, 55, 66, 78, 91, 105, 120, 136, 153, 171, 190, 210,
 231, 253, 276, 300, 325, 351, 378, 406, 435, 465, 496, 528, 561, 595, 630, 666,
 703, 741, 780, 820, 861, 903, 946, 990, 1035, 1081, 1128, 1176, 1225, 1275}
```

```
OEIS[tempdata, Infinity]
```

OEIS Query: {0, 1, 3, 6, 10, 15, 21, 28, 36, 45, 55, 66, 78, 91, 105, 120, 136, 153, 171, 190, 210, 231, 253, 276, 300, 325, 351, 378, 406, 435, 465, 496, 528, 561, 595, 630, 666, 703, 741, 780, 820, 861, 903, 946, 990, 1035, 1081, 1128, 1176, 1225, 1275}

The On-Line Encyclopedia of Integer Sequences, published electronically at http://oeis.org, 2010

Go to OEIS complete search results

Summary display of results 1-5 out of 5 results found.

{{A000217, Triangular numbers: a(n) = C(n+ 1 ,2) = n(n+ 1)/2 = 0 + 1 +2+...+n. (Formerly M2535 N1002) , +1020 1705 }, {**0**, **1**, **3**, **6**, **10**, **15**, **21**, **28**, **36**, **45**, **55**, **66**, **78**, **91**, **105**, **120**, **136**, **153**, **171**, **190**, **210**, **231**, **253**, **276**, **300**, **325**, **351**, **378**, **406**, **435**, **465**, **496**, **528**, **561**, **595**, **630**, **666**, **703**, **741**, **780**, **820**, **861**, **903**, **946**, **990**, **1035**, **1081**, **1128**, **1176**, **1225**, **1275**, 1326, 1378, 1431}}

{{A105340, a(n) = n*(n+ 1)/2 mod 2048., +1020 1 }, {**0**, **1**, **3**, **6**, **10**, **15**, **21**, **28**, **36**, **45**, **55**, **66**, **78**, **91**, **105**, **120**, **136**, **153**, **171**, **190**, **210**, **231**, **253**, **276**, **300**, **325**, **351**, **378**, **406**, **435**, **465**, **496**, **528**, **561**, **595**, **630**, **666**, **703**, **741**, **780**, **820**, **861**, **903**, **946**, **990**, **1035**, **1081**, **1128**, **1176**, **1225**, **1275**, 1326, 1378, 1431, 1485}}

{{A161680, Cumulative frequency distribution of numbers in A003057 ., +1020 1 }, {**0**, **0**, **1**, **3**, **6**, **10**, **15**, **21**, **28**, **36**, **45**, **55**, **66**, **78**, **91**, **105**, **120**, **136**, **153**, **171**, **190**, **210**, **231**, **253**, **276**, **300**, **325**, **351**, **378**, **406**, **435**, **465**, **496**, **528**, **561**, **595**, **630**, **666**, **703**, **741**, **780**, **820**, **861**, **903**, **946**, **990**, **1035**, **1081**, **1128**, **1176**, **1225**, **1275**}}

{{A089594, Alternating sum of squares to n., +1010 0 }, {-1, **3**, -6, **10**, -15, **21**, -28, **36**, -45, **55**, -66, **78**, -91, **105**, -120, **136**, -153, **171**, -190, **210**, -231, **253**, -276, **300**, -325, **351**, -378, **406**, -435, **465**, -496, **528**, -561, **595**, -630, **666**, -703, **741**, -780, **820**, -861, **903**, -946, **990**, -1035, **1081**, -1128, **1176**, -1225, **1275**}}

{{A132654, An 8 X 8 magic square with consecutive triangular numbers, read by rows., +1010 0 }, {1596, **595**, **36**, 1653, **171**, **1128**, **45**, **496**, **561**, **210**, 1485, **1176**, **28**, **435**, 1770, **55**, **351**, **946**, **91**, **276**, 2080, **741**, **10**, **1225**, **190**, **15**, **630**, **465**, 1431, **78**, **1081**, 1830, **120**, **325**, 2016, **3**, **861**, **300**, **1275**, **820**, **21**, 1540, **153**, **66**, **666**, 1711, **528**, **1035**, 1891, **136**, **903**, 1378, **378**, **1**, **780**, **253**, **990**, 1953, **406**, **703**, **105**, 1326, **231**, **6**}}

3. Offsets

■ Mathematica

1. Not open source

2. FindSequenceFunction: sometimes gives ‘incorrect’ formulas (sensitive to number of terms used)

High-Performance Computing

1. Sequences with extremely large integers

2. Large number of searches: perform search using OEIS website or download OEIS content (label, name, sequence, offset)

3. Each entry generates 47 sequences that need to be searched for in OEIS database (6 subsequences, 8 transformations)

4. My PC (Dell Latitude D630) can mine approximately 500 entries per day (running continuously)

5. Over 8 million searches are needed to mine all entries in OEIS database (over 180,000); this requires running a single PC continuously for one year.

6. Memory intensive: OEIS database (50 MB), storage of results

False Positives in Matching Sequences

****

A004529: Ratios of successive terms are 1,1,1,2,3,3,3,4,5,5,5,6...

{a[n]}={1, 1, 1, 1, 2, 6, 18, 54, 216, 1080, 5400, 27 000, 162 000, 1 134 000, 7 938 000, 55 566 000, 444 528 000, 4 000 752 000, 36 006 768 000, 324 060 912 000, 3 240 609 120 000, 35 646 700 320 000, 392 113 703 520 000, 4 313 250 738 720 000}

$\{a[2^n]\}=\{1, 1, 2, 216, 444528000\}$

$\sum_{k=0}^{n} a\left[2^k\right]^2 =$ A135408 a(1)=1. a(n) = a(n-1) + a(n-1)^a(n-1).

****

****

Conjecture 316:

a[n]=A001441 Number of inequivalent Costas arrays of order n under dihedral group.

b[n]=A002013 Filaments with n square cells. (Formerly M0835 N0317)

c[n]=A003820 a(1)=a(2)=1, a(n+1) = (a(n)^5 +1)/a(n-1).

$\sum_{k=0}^{n} a\left[2^k\right] \text{Binomial}[n, k] = \sum_{k=0}^{n} b\left[2^k\right] = \sum_{k=0}^{n} c[k]\, c[-k+n] =$ A175169

Numbers n such that n divides the sum of digits of 2^n.

****

****

A006144: Number of self-avoiding walks on square lattice. (Formerly M3242)

{a[n]}={0, 1, 0, 0, 0, 4, 5, 6, 11, 31, 72, 157, 312, 700, 1472, 3446, 7855}

$\{a[2^n]\}=\{0, 0, 0, 0, 312\}$

$a[2^n] =$ A022066 Theta series of D*_13 lattice.

****

****

Conjecture 430:

a[n]=A005708 a(n)=a(n-1)+a(n-6). (Formerly M0496)

b[n]=A005840 Expansion of (1-x)*e^x/(2-e^x). (Formerly M1872)

$\sum_{k=0}^{n} a\left[k^2\right] \text{Binomial}[n, k] = \sum_{k=0}^{n} b[k] =$ A177921

Number of oval-partitions of the regular n-gon {2n}.

****

EUREKA *Mathematica* Package Version 0.5

■ Sequence Matching Algorithm




No match
found
One match
found
≥2 matches
found
i++.
k(i)=k(i−1)−Δ
Full match?
i++.
k(i)=k(i−1)+Δ
YES.
Unique match found.
Stop
NO.
Stop.

■ Sample *Mathematica* code

```
While[(toggleO == 0 && step < stepMax && lengthdatamiddle > 4) ||
   (toggleO == 0 && lengthOEIS > 1 && step < stepMax),

    lengthdatamiddle = Ceiling[(lengthdatatop + lengthdatabottom) / 2];
    If[lengthdatamiddle > 4 || (lengthdatamiddle <= 4 && lengthOEIS < 1),

        If[lengthdatamiddle <= 4, lengthdatamiddle = 4
        ];

        matchOEIS = OEISMatchTwo[Take[tempdata, lengthdatamiddle]];
        lengthOEIS = Length[matchOEIS];

        If[lengthOEIS == 1,
            toggleO = 1
        ];

(* If no match at all, then consider negative of sequence;
    if still no match, then decrease number of terms *)

        If[lengthOEIS < 1,
            matchOEISsign = OEISMatchTwo[Take[-tempdata, lengthdatamiddle]];

            If[Length[matchOEISsign] >= 1,

                tempdata = -tempdata;
                matchOEIS = matchOEISsign;
                lengthOEIS = Length[matchOEIS],

                lengthdatatop = lengthdatamiddle
            ],

            matchOEISsign = OEISMatchTwo[Take[-tempdata, lengthdatamiddle]];

(* If more than one match, then increase number of terms*)

            If[lengthOEIS > 1,

                If[Length[matchOEISsign] == 1 && step == stepMax - 1,

                    toggleO = 1; tempdata = -tempdata; matchOEIS = matchOEISsign,

                    status = lengthdatamiddle;
                    statusOEIS = matchOEIS;
                    lengthstatusOEIS = Length[statusOEIS];
```

```
                    lengthdatabottom = lengthdatamiddle
                ],

    (* If unique match, then confirm full match *)

                toggleO = 1;
                If[Length[matchOEISsign] == 1,
                    matchOEISdata = OEISDatabaseData[[matchOEIS[[1]]]];
                    matchOEISsigndata = OEISDatabaseData[[matchOEISsign[[1]]]];
                    position = Position[matchOEISdata, tempdata[[1]]];

        positionsign = Position[matchOEISsigndata, (-tempdata)[[1]]];
                    If[position[[1, 1]] > positionsign[[1, 1]],
         tempdata = -tempdata;
                        matchOEIS = matchOEISsign
                    ]
                ]
            ];

            If[lengthOEIS > 1 && lengthdatamiddle == 4,
                statusOEIS = matchOEIS; status = lengthdatamiddle
            ]
        ]
    ]
    step++;
];
```

Sample output

```
OEISIdentitySearch["A000041", "A000041", {1, 6}, {1, 8}]
```

```
*****************************************************
A000041
Using Mathematica formula to extrapolate a[n] to about 100 terms:
A000041:
 a(n) = number of partitions of n (the partition numbers). (Formerly M0663 N0244)
{a[n]}={1, 1, 2, 3, 5, 7, 11, 15, 22, 30, 42, 56, 77, 101, 135,
  176, 231, <<66>>, 23 338 469, 26 543 660, 30 167 357, 34 262 962, 38 887 673,
  44 108 109, 49 995 925, 56 634 173, 64 112 359, 72 533 807, 82 010 177,
  92 669 720, 104 651 419, 118 114 304, 133 230 930, 150 198 136, 169 229 875}
*****************************************************
RUN 1
```

{a[n]}={1, 1, 2, 3, 5, 7, 11, 15, 22, 30, 42, 56, 77, 101, 135,
176, 231, <<66>>, 23 338 469, 26 543 660, 30 167 357, 34 262 962, 38 887 673,
44 108 109, 49 995 925, 56 634 173, 64 112 359, 72 533 807, 82 010 177,
92 669 720, 104 651 419, 118 114 304, 133 230 930, 150 198 136, 169 229 875}

Eureka!

OEIS Formula Found:

$\sum_{k=0}^{n} a[k] =$ A000070 Sum_{k=0..n} p(k) where p(k) =
number of partitions of k (A000041). (Formerly M1054 N0396)

OEIS Multiple Partial Matches Found: LCS={1, 2, 6, 15, 40}

$\sum_{k=0}^{n} a[k]^2 =$ Multiple partial matches found

OEIS Longest Partial Match: LCS={1, 0, 1, 1, 2, 4, 9, 21}

$\sum_{k=0}^{n} (-1)^k a[k]$ Binomial[n, k] = A168049 Expansion of (3-x-sqrt(1-2x-3x^2))/2.

Eureka!

OEIS Formula Found:

$\sum_{k=0}^{n} a[k]\, a[-k+n] =$ A000712
Number of partitions of n into parts of 2 kinds. (Formerly M1376 N0536)

Eureka!

OEIS Formula Found:

$\sum_{k=0}^{n} k\, a[k] =$ A141156 Row sums of triangle A141155 .

OEIS Multiple Partial Matches Found: LCS={1, 2, 5, 13, 34, 88}

$\sum_{k=0}^{n} a[k]$ Binomial[n, k] = Multiple partial matches found

----

Eureka!

OEIS Formula Found:

a[n] a[1 + n] = A090982 Partitions(n)*Partitions(n+1).

****

RUN 2

{a[2 n]} = {1, 2, 5, 11, 22, 42, 77, 135, 231, 385, 627, 1002, 1575, 2436, 3718, 5604, <<18>>, 3 087 735, 4 087 968, 5 392 783, 7 089 500, 9 289 091, 12 132 164, 15 796 476, 20 506 255, 26 543 660, 34 262 962, 44 108 109, 56 634 173, 72 533 807, 92 669 720, 118 114 304, 150 198 136}

Eureka!

OEIS Formula Found:

a[2 n] = A058696 Number of ways to partition 2n into positive integers.

OEIS Multiple Partial Matches Found: LCS={1, 3, 8, 19, 41}

$\sum_{k=0}^{n} a[2k]$ = Multiple partial matches found

OEIS Longest Partial Match: LCS={1, 5, 30, 151}

$\sum_{k=0}^{n} a[2k]^2$ = A055298 Number of trees with n nodes and 11 leaves.

OEIS Longest Partial Match: LCS={1, -1, 2, -1, 1, -1, -1}

$\sum_{k=0}^{n} (-1)^k a[2k]$ Binomial[n, k] = A115413 G.f.: (x - 1)/(1 - x^2 + x^3 + x^4 - x^5).

OEIS Longest Partial Match: LCS={1, 4, 14, 42, 113}

$\sum_{k=0}^{n} a[2k]\, a[2(-k+n)]$ = A124616 Poincare series P(T_{4,2}; x).

OEIS Multiple Partial Matches Found: LCS={1, 3, 10, 33, 105}

$\sum_{k=0}^{n} a[2k]$ Binomial[n, k] = Multiple partial matches found

****

RUN 3

{a[1 + 2 n]} =
{1, 3, 7, 15, 30, 56, 101, 176, 297, 490, 792, 1255, 1958, 3010, 4565, 6842, <<18>>,
3 554 345, 4 697 205, 6 185 689, 8 118 264, 10 619 863, 13 848 650, 18 004 327, 23 338 469,
30 167 357, 38 887 673, 49 995 925, 64 112 359, 82 010 177, 104 651 419, 133 230 930, 169 229 875}

Eureka!

OEIS Formula Found:

a[1 + 2 n] = A058695 Number of ways to partition 2n+1 into positive integers.

OEIS Longest Partial Match: LCS={1, 4, 11, 26, 56, 112}

$\sum_{k=0}^{n}$ a[1 + 2 k] = A027660 C(n+2,2)+C(n+2,3)+C(n+2,4)+C(n+2,5).

OEIS Multiple Partial Matches Found: LCS={1, -2, 2, -2, 1, 0}

$\sum_{k=0}^{n} (-1)^k$ a[1 + 2 k] Binomial[n, k] = Multiple partial matches found

----

OEIS Longest Partial Match: LCS={1, 6, 23, 72}

$\sum_{k=0}^{n}$ a[1 + 2 k] a[1 + 2 (-k + n)] = A045618 Partial sums of A000337 (n+4), n >= 0.

OEIS Longest Partial Match: LCS={1, 7, 28, 88}

$\sum_{k=0}^{n}$ k a[1 + 2 k] = A163037

Number of nX2 binary arrays with all 1s connected and a path of
1s from left column to right column

OEIS Multiple Partial Matches Found: LCS={1, 4, 14, 46, 145}

$\sum_{k=0}^{n}$ a[1 + 2 k] Binomial[n, k] = Multiple partial matches found

****

RUN 4

{a[n^2]} = {1, 1, 5, 30, 231, 1958, 17 977, 173 525, 1 741 630, 18 004 327}

Eureka!

OEIS Formula Found:

$a[n^2] =$ A072213 Number of partitions of n^2.

OEIS Longest Partial Match: LCS={1, 2, 7, 37, 268}

$\sum_{k=0}^{n} a[k^2] =$ A107877 Column 1 of triangle A107876 .

OEIS Longest Partial Match: LCS={1, 2, 11, 70}

$\sum_{k=0}^{n} a[k^2]\, a[(-k+n)^2] =$ A118347

Semi-diagonal (one row below central terms) of pendular triangle A118345 and equal to the self-convolution of the central terms (A118346).

OEIS Longest Partial Match: LCS={3, 18, 138}

$\sum_{k=0}^{n} k\, a[k^2] =$ A039618

Number of 2n-step self-avoiding closed walks on first octant of 3-dimensional cubic lattice, passing through origin.

OEIS Multiple Partial Matches Found: LCS={1, 2, 8, 49}

$\sum_{k=0}^{n} a[k^2]$ Binomial[n, k] = Multiple partial matches found

****

RUN 5

$\{a[2^n]\}=\{1, 2, 5, 22, 231, 8349, 1741630\}$

Eureka!

OEIS Formula Found:

$a[2^n] =$ A068413 a(n) = number of partitions of 2^n.

OEIS Multiple Partial Matches Found: LCS={1, 3, 8, 30}

$\sum_{k=0}^{n} a[2^k] =$ Multiple partial matches found

OEIS Multiple Partial Matches Found: LCS={1, 4, 14, 64}

$\sum_{k=0}^{n} a[2^k]\, a[2^{-k+n}]$ = Multiple partial matches found

----

OEIS Multiple Partial Matches Found: LCS={1, 3, 10, 44}

$\sum_{k=0}^{n} a[2^k]$ Binomial[n, k] = Multiple partial matches found

****

RUN 6

{a[Prime[n]]}={2, 3, 7, 15, 56, 101, 297, 490, 1255, 4565,
6842, 21 637, 44 583, 63 261, 124 754, 329 931, 831 820, 1 121 505, 2 679 689,
4 697 205, 6 185 689, 13 848 650, 23 338 469, 49 995 925, 133 230 930}

Eureka!

OEIS Formula Found:

a[Prime[n]] = A058698

p(P(n)), P = primes (A000040), p = partition numbers (A000041).

OEIS Multiple Partial Matches Found: LCS={2, 5, 12, 27}

$\sum_{k=0}^{n}$ a[Prime[k]] = Multiple partial matches found

----

OEIS Longest Partial Match: LCS={13, 62, 287}

$\sum_{k=0}^{n}$ a[Prime[k]]^2 = A141786

Counts of Kekulean pericondensed planar benzenoid hydrocarbons
(see reference for precise definition).

OEIS Multiple Partial Matches Found: LCS={-2, -1, -4, -3}

$\sum_{k=0}^{n} (-1)^k$ a[Prime[k]] Binomial[n, k] = Multiple partial matches found

----

OEIS Multiple Partial Matches Found: LCS={2, 7, 22, 69}

$\sum_{k=0}^{n}$ a[Prime[k]] Binomial[n, k] = Multiple partial matches found

****

End of search.

9 OEIS formulas found for A000041 (saved to identitiesA000041-A000041.txt).

38 new unrecognized sequences found (saved to OEISNewEntriesA000041-A000041.txt).

Statistics

- **10,000 entries mined so far using 8 different transformations, 6 subsequences (with many bugs along the way)**
- **1.5 months run-time on a laptop PC (Dell Latitude D630)**
- **3860 “formulas” found (unique matches recognized by OEIS) - 3.09 MB file**
- **590 “identities” found (experimental conjectures). Preliminary analysis shows:**
  - Most identities are trivial or already mentioned in OEIS (>90%)
  - Small fraction of unrecognized identities (further analysis required) (<5%)
  - Small fraction of false positives (<5%)
- **290,406 new sequences generated (unrecognized by OEIS) - 51.3 MB file (Unmined)**

A Sample of Experimental Conjectures by Eureka

Example 1

****

Conjecture 4:

a[n]=A000032 Lucas numbers (beginning at 2): L(n) = L(n-1) + L(n-2). (Cf. A000204 .) (Formerly M0155)

b[n]=A000204 Lucas numbers (beginning with 1): L(n) = L(n-1) + L(n-2) with L(1) = 1, L(2) = 3. (Formerly M2341 N0924)

c[n]=A002715
An infinite coprime sequence defined by recursion. (Formerly M2683 N1073)

d[n]=A005247
a(n) = 3a(n-2) - a(n-4), a(0)=2, a(1)=1, a(2)=3, a(3)=2. Alternates Lucas (A000032) and Fibonacci (A000045) sequences for even and odd n. (Formerly M0149)

e[n]=A005248
Bisection of Lucas numbers: a(n) = L(2n) = A000032 (2n). (Formerly M0848)

$a[2^n]=b[2^n]=c[1+2n]=d[2^n]=e[2^n]$=A001566
a(0) = 3; thereafter, a(n) = a(n-1)^2 - 2. (Formerly M2705 N1084)

****

Example 2

****

Conjecture 105:

a[n]=A000211 a(n) = a(n-1) + a(n-2) - 2. (Formerly M2396 N0953)

b[n]=A001254 Squares of Lucas numbers.

$a[2^n]=b[2^n]$=A000324
A nonlinear recurrence: a(n) = a(n-1)^2-4*a(n-1)+4 (for n>1). (Formerly M3789 N1544)

****

Example 3

****

Conjecture 208:

a[n]=A000740
Number of 2n-bead balanced binary necklaces of fundamental period 2n, equivalent to reversed complement; also Dirichlet convolution of b_n=2^(n-1) with mu(n); also number of components of Mandelbrot set corresponding to Julia sets with an attractive n-cycle. (Formerly M2582 N1021)

b[n]=A003465 Number of ways to cover an n-set. (Formerly M4024)

c[n]=A003473 Generalized Euler PHI function. (Formerly M0875)

d[n]=A004730 Numerator of n!!/(n+1)!!.

e[n]=A004732 Numerator of n!!/(n+3)!!.

$\sum_{k=0}^{n} a[2^k] = \sum_{k=0}^{n} b[k]\,\text{Binomial}[n, k] = c[2^n] = d[2^n] = e[2^n] =$ A058891 2^(2^(n-1)-1).

****

Example 4

****

Conjecture 395:

a[n]=A004011 Theta series of D_4 lattice; Fourier coefficients of Eisenstein series E_{gamma,2}. (Formerly M5140)

$\sum_{k=0}^{n} a[k] = \sum_{k=0}^{n} a[2k] =$ A046949 Sizes of successive balls in D_4 lattice.

****

Example 5

****

Conjecture 396:

a[n]=A004187 a(n) = 7*a(n-1) - a(n-2) with a(0) = 0, a(1) = 1.

$a[n]\,a[1+n] = \sum_{k=0}^{n} a[2k] =$ A161582

The list of the k values in the common solutions to the 2 equations 5*k+1=A^2, 9*k+1=B^2.

****

Example 6

****

Conjecture 398:

a[n]=A004254 a(n) = 5a(n - 1) - a(n - 2), a(0) = 0, a(1) = 1. (Formerly M3930)

$a[n]\ a[1+n] = \sum_{k=0}^{n} a[2\,k]$ =A160695

a(n) such that 3*a(n)+1 and 7*a(n)+1 are both perfect squares.

****

Example 7

****

Conjecture 427:

a[n]=A005251 a(0) = 0, a(1) = a(2) = a(3) =
1; thereafter, a(n) = a(n-1)+a(n-2)+a(n-4). (Formerly M1059)

b[n]=A005314 For n = 0, 1, 2, a(n) = n;
thereafter, a(n) = 2a(n-1)-a(n-2)+a(n-3). (Formerly M0709)

$\sum_{k=0}^{n} a[2\,k]\ \mathrm{Binomial}[n, k] = \sum_{k=0}^{n} b[1+2\,k]\ \mathrm{Binomial}[n, k]$

=A012781 Take every 5th term of Padovan sequence A000931 .

****

Next Steps

Scale up processing power and memory

- **Need faster computers, more memory**
- **Integrate parallel computing: multi-core CPU’s, multiple CPU’s, cluster computing**

Improve search algorithms

- **Reduce run-times**
- **Reduce false positives**

Expand Scope of Search

- **Increase bank of sequence transformations**
- **Data mine collection of new (unrecognized) sequences generated**
- **Extend algorithms to 2-D sequences, rational sequences (e.g. Bernoulli numbers)**

Disseminate Work

- **Create database website**

Seek Help

- **Need editors to analyze EUREKA’s conjectures: filter out trivial conjectures and false positives**
- **Need good programmers (recruit students!)**

The End